

## Controlling AmigaTeX from CygnusEd

Garry Glendown

One of the many fine features that Tom Rokicki has added to AmigaTeX is the possibility of controlling the main TeX commands via an ARexx interface. So why not use that?

All you need is a version of AmigaTeX, Cygnus-Ed Professional, AmigaREXX, and lots of memory (1.5 Mb is enough). From CLI, start CED, preview, and finally TeX. As the first two detach totally, you won't need any RUNs.

Now enter one of the two listings below. But before you do, check for the version of your AmigaTeX release. If you have anything since 2.9m, use the first listing. Before that, use the second one. Store it somewhere on your hard disk or disk drive. Now install it as a REXX command using the "SPECIAL.Dos/ARexx interface.install Dos/ARexx command" function.

When using any TeX file, you may now press the function key you've assigned to that program, and CED will run TeX. If there are any errors, it will try to jump to the position returned by AmigaTeX. And, if you forgot the '\end' or '\bye' at the end of your file, '\end' will automatically be inserted at the end for you.

**Bugs.** I've had some problems with the old AmigaTeX versions, as they would sometimes do weird things when being called with ARexx. But with the new one (2.9m) and the new REXX program, everything has gone fine the last couple months. So if you still have the old AmigaTeX version, contact your dealer and get the new one (especially with the AmigaTeX-version with IFF-support being out...)!

**Disclaimer.** I'm not bad in C, 68000 Assembler and TeX, but I'm terrible in REXX. I was happy when the old version finally worked, and was as disappointed when it didn't work with 2.9m. And again, I was happy when the new version worked with 2.9m. I didn't care too much about how the code looked, but I'll accept all hints and remarks about it with a thankful smile. Just forward 'em to the address at the end of the listings.

```

/*****
* TeXify.ced - Tells TeX about the new file
*               and compiles it
*               for AmigaTeX Version >= 2.9m
* (C) 1989 by G.Glendown
*
* This is the version for newly distributed
* version of AmigaTeX. The new command

```

```

* 'NextPrompt' helped reduce the code quite
* a bit, but I'm still not too happy with
* it. If you have any trouble or find bugs,
* send a note to:
*               Garry@DGIHRZ01.Bitnet
*****/

options failat 5
options results
address 'rexx_ced'
status 21
srcfile=result
sfxlgt=length(srcfile)-lastpos('.',srcfile)
sfx=upper(right(srcfile,sfxlgt))
if sfx~='TEX' then do /* does the filename
                        have a '.tex' ? */
    okay1 'Sorry, not a TeX-Sourcefile!'
    exit
end
status 18 /* get number of changes since
            last save */
chnge=result
if chng~=0 then do /* if chng~=0 then save */
    'save'
end
file=left(srcfile,lastpos('.',srcfile)-1)
address 'AmigaTeX'
'ToFront'
'TeXify ' file
'NextPrompt' /* I've had problems with
                one NextPrompt, */
'NextPrompt' /* so I use two... */
'Prompt'
prompt = getclip('AmigaTeX.Prompt')
'ToBack'
if prompt='**' then do
                /* compiling finished */
    address 'rexx_ced'
    exit
end
if prompt='*' then do /* no \end */
    'TeXify \end'
    address 'rexx_ced'
    'Jumpto' 99999 99999
    'text' 'OA'X
    'text' '\end' 'OA'X
    address 'AmigaTeX' 'Abort'
    address 'rexx_ced'
    okay1 '\end inserted!'
    exit
end
'Abort'
'ErrorLoc'
error=getclip('AmigaTeX.ErrorLoc')

```

```
'ToFront'      /* Get the CLI-Screen to
                the front */
'ToBack'       /* and put the CLI with
                TeX to the back again */
address 'rexx_ced'
okay1 'Guess what! I found an error! '
'expand view'
say error
parse var error name line pos
'Jump to' line pos+1
```

For releases of AmigaTeX before 2.9m:

```
/******
 * TeXify.ced - Tells TeX about the new file
 *               and compiles it
 *               for AmigaTeX w/o new REXX-Commands
 * (C) 1989 by G.Glendown
 *
 * I'm not too good at REXX yet, so this
 * program is far off being perfect. I've had
 * some problems with deadlocks which lead me
 * to put that many 'Prompt'-commands in.
 * I've reduced the deadlocks to 'once in a
 * while', but they still occur. If you find
 * out where the problem is, please tell me:
 * Garry@DGIHRZ01.Bitnet
 *****/
```

```
options failat 5
options results
address 'rexx_ced'
status 21
srcfile=result
sfxlgt=length(srcfile)-lastpos('.',srcfile)
sfx=upper(right(srcfile,sfxlgt))
if sfx~='TEX' then do /* does the filename
                    have a '.tex' ? */
    okay1 'Sorry, not a TeX-Sourcefile!'
    exit
end
status 18 /* get number of changes since
          last save */
chnge=result
if chnge~=0 then do /* if chnge~=0 then save */
    'save'
end
file=left(srcfile,lastpos('.',srcfile)-1)
address 'AmigaTeX'
prompt=''
do until prompt='**'
    'Prompt'
    prompt = getclip('AmigaTeX.Prompt')
end
```

```
prompt=''
'TeXify' file
do until hash(prompt)=0
    'Prompt'
    prompt = getclip('AmigaTeX.Prompt')
end

prompt=''
do until hash(prompt)~=0
    'Prompt'
    prompt = getclip('AmigaTeX.Prompt')
end

prompt=''
'Prompt'
prompt = getclip('AmigaTeX.Prompt')
'ToBack'
if prompt='**' then do
    /* compiling finished */
    address 'rexx_ced'
    exit
end
if prompt='*' then do /* no \end */
    'TeXify \end'
    address 'rexx_ced'
    'Jump to' 99999 99999
    'text' 'OA'X
    'text' '\end' 'OA'X
    address 'AmigaTeX' 'Abort'
    address 'rexx_ced'
    okay1 '\end inserted!'
    exit
end
'Abort'
'ErrorLoc'
error=getclip('AmigaTeX.ErrorLoc')
address 'rexx_ced'
okay1 'Guess what! I found an error! '
'expand view'
say error
parse var error name line pos
'Jump to' line pos+1
```

◇ Garry Glendown  
Box R, APO NY 09141  
or  
Guldene Kammer 35  
6430 Bad Hersfeld  
FRG  
Tel.: 0662177923  
Bitnet: Garry@DGIHRZ01